

Sorting in OCaml

Sorting algorithms implemented in the OCaml programming language

Clément Pit--Claudel

March 27, 2010

Abstract

In this paper, we present implementations in the OCaml programming language of many popular sorting algorithms, with complexities ranging from quadratic ($O(n^2)$) to logarithmic ($O(n \lg n)$). We then conduct a few tests, in order to evaluate the relative sorting performance of these algorithms. Finally, through careful timing, we demonstrate that our supposedly unoptimized algorithms achieve an approximate 30% performance increase over the OCaml standard library sorting functions.

Keywords: OCaml, Caml Light, Complexity theory, Sorting algorithms, Bubble sort, Insert sort, Selection sort, Merge sort, Quicksort, Heap sort

Introduction The OCaml programming language, despite providing programmers with lots of useful, high-level constructs which drastically reduce development time, is still much of a work-in-progress. Its performance is overall bad, especially when running on Windows systems, and the measurements we present here are a great illustration of how much remains to be done. When designing the algorithms used in this article, we preferred focusing on simplicity rather than optimization. They do, however, demonstrate excellent performance.

1 Base functions and definitions

```

1 let swap a b t =
2   let t_b = t.(b) in t.(b) <- t.(a); t.(a) <- t_b;;

```

2 Quadratic sorting algorithms

The following algorithms all have $O(n^2)$ time complexity. Space complexity is in all cases $O(n)$. Recursive algorithms (operating on lists) are denoted by a prefixing “r”.

2.1 Bubble sort

```

1 let bubble_sort t =
2   let length = Array.length t in
3   for max_pos = length - 1 downto 0 do
4     for pos = 0 to max_pos - 1 do
5       if t.(pos) > t.(pos + 1) then swap pos (pos + 1) t
6     done
7   done
8 ;;

1 let rec r_bubble_sort l =
2   let rec try_swap = function
3     | [] -> (false, [])
4     | [a] -> (false, [a])
5     | a :: b :: tail ->
6       if a > b then (true, b :: (snd (try_swap (a :: tail))))
7       else let swapped, newlist = try_swap (b :: tail) in (swapped, a ::
8         newlist)
9   in let rec sort l =
10     let swapped, newlist = try_swap l in
11     if swapped then sort newlist else newlist
12   in sort l
13 ;;

```

2.2 Insert sort

```

1 let insert_sort t =
2   let length = Array.length t in
3   for start = 1 to length - 1 do
4     let pos = ref start in
5     while !pos > 0 && t.(!pos - 1) > t.(!pos) do
6       swap (!pos-1) (!pos) t;
7       decr pos
8     done
9   done
10 ;;

1 let r_insert_sort l =
2   let rec insert value = function
3     | [] -> [value]
4     | h :: t ->
5       if h > value then value :: h :: t
6       else h :: insert value t
7   in let rec sort = function
8     | [] -> []
9     | h :: t -> insert h (sort t)
10   in sort l
11 ;;

```

2.3 Selection sort

```

1  let selection_sort t =
2    let length = Array.length t in
3    for current = 0 to length - 1 do
4      let min_id = ref current in
5      for pos = current to length - 1 do
6        min_id := if (t.(pos) < t.(!min_id)) then pos else !min_id
7      done;
8      swap current !min_id t;
9    done
10   ;;

1  let rec r_selection_sort l =
2    let rec get_min = function
3      | [] -> failwith "Empty"
4      | [m] -> m
5      | h :: t -> min h (get_min t)
6    in let rec rem_min m = function
7      | [] -> failwith "Empty"
8      | h :: t -> if h = m then t else h :: rem_min m t
9    in let rec sort = function
10     | [] -> []
11     | liste ->
12       let min_val = get_min liste in
13       min_val :: (sort (rem_min min_val liste))
14   in sort l
15   ;;

1  let rec r_selection_sort2 l =
2    let rec extract = function
3      | [] -> failwith "Empty"
4      | [a] -> (a, [])
5      | head :: tail ->
6        let tail_min, newtail = extract tail in
7        if tail_min > head then (head, tail_min :: newtail)
8        else (tail_min, head :: newtail)
9    in let rec sort = function
10     | [] -> []
11     | liste -> let min, tail = extract liste in min :: sort tail
12   in sort l
13   ;;

```

Note For performance reason, we provide two implementations of the recursive selection sort. The first version, although less subtle, demonstrates better performance.

3 Logarithmic sorting algorithms

The following algorithms all have $O(n \log(n))$ average time complexity, although quicksort has a worst-case performance of $O(n^2)$ (the others having a worst case of $O(n \log(n))$). Space complexity varies, although usually being $O(n)$. Note that only quicksort is an in-place algorithm.

3.1 Merge sort

```

1  let merge_sort t =
2    let merge t1 length1 t2 length2 =
3      let out = Array.make (length1 + length2) 0 in
4      let pos1, pos2 = ref 0, ref 0 in
5      while !pos1 < length1 || !pos2 < length2 do
6        let pos = !pos1 + !pos2 in
7        if !pos1 = length1 then
8          (out.(pos) <- t2.(!pos2); incr pos2)
9        else if !pos2 = length2 then
10         (out.(pos) <- t1.(!pos1); incr pos1)
11        else
12         begin
13           if t1.(!pos1) < t2.(!pos2) then
14             (out.(pos) <- t1.(!pos1); incr pos1)
15           else
16             (out.(pos) <- t2.(!pos2); incr pos2)
17         end;
18      done; out;
19  in
20
21  let rec sort start length =
22    if length = 0 then []|]
23    else if length = 1 then [t.(start)]|]
24    else
25      begin
26        let start1 = start and length1 = length / 2 in
27        let start2 = start1 + length1 and length2 = length - length1 in
28        merge (sort start1 length1) length1 (sort start2 length2) length2
29      end
30  in
31  let size = Array.length t in Array.blit (sort 0 size) 0 t 0 size
32  ;;

```

For the recursive version, we define a `split` function as follows:

```

1  let split l =
2    let rec _split source left right =
3      match source, right with
4      | [], _ | [_], _ -> (left, right)
5      | _ :: _ :: tail, r :: right_tail -> _split tail (r :: left)
6        right_tail
7      | _ -> assert false
9    in _split l [] l;;

```

Note The `split` function can also be implemented as follows (a little slower though):

```

1  let split l =
2    let rec _split source left right =
3      match source with
4      | [] -> (left, right)
5      | head :: tail -> _split tail right (head :: left)
6    in _split l [] [];;

```

```

1  let r_merge_sort l =
2    let rec merge l1 l2 =
3      match l1, l2 with
4      | [], l | l, [] -> l
5      | h1 :: t1, h2 :: t2 ->
6          if h1 < h2 then h1 :: (merge t1 (h2 :: t2))
7          else             h2 :: (merge (h1 :: t1) t2)
8    in
9
10   let rec sort = function
11     | ([] | [_]) as sorted -> sorted
12     | list -> let left, right = split list in merge (sort left) (sort right)
13   in sort l
14 ;;

```

```

1  let tr_merge_sort l =
2    let merge l1 l2 =
3      let rec _merge l1 l2 result =
4        match l1, l2 with
5        | [], [] -> result
6        | [], h :: t | h :: t, [] -> _merge [] t (h :: result)
7        | h1 :: t1, h2 :: t2 ->
8            if h1 < h2 then _merge t1 l2 (h1 :: result)
9            else             _merge l1 t2 (h2 :: result)
10       in List.rev (_merge l1 l2 [])
11   in
12
13   let rec sort l merge_fn =
14     match l with
15     | [] | [_] -> merge_fn l
16     | list -> let left, right = split list in
17                 sort left (fun leftR -> sort right (fun rightR ->
18                     merge_fn (merge leftR rightR)))
19   in sort l (fun x -> x)

```

3.2 Quick sort

```

1  let quicksort t =
2    let split start length pivot_pos =
3      let pivot = t.(pivot_pos) in
4        swap start pivot_pos t;
5
6      let low, high = ref (start + 1), ref (start + length - 1) in
7      while !low < !high do
8        while !low < !high && t.(!low) <= pivot do
9          incr low
10         done;
11
12        while !low < !high && t.(!high) >= pivot do
13          decr high
14         done;
15
16        if !low < !high then swap !low !high t
17       done;
18       if t.(!low) > pivot then decr low;
19       swap start !low t;
20       !low;
21     in
22
23     let rec sort start length =
24       if length > 1 then
25         begin
26           let pivot_pos = start + Random.int length in
27           let new_pos = split start length pivot_pos in
28             sort start (new_pos - start);
29             sort (new_pos + 1) (start + length - new_pos - 1);
30         end
31     in sort 0 (Array.length t)
32 ;;

```

```

1  let r_quicksort l =
2    let split list pivot =
3      let rec _split inf sup = function
4        | [] -> (inf, sup)
5        | h :: t ->
6          if h < pivot then _split (h :: inf) sup t
7          else _split inf (h :: sup) t
8      in _split [] [] list
9    in
10
11    let rec sort result = function
12      | [] -> result
13      | [a] -> a :: result
14      | pivot :: t -> let (inf, sup) = split t pivot in sort (pivot :: (sort
15        result inf)) sup
16    in List.rev (sort [] l)
17 ;;

```

3.3 Heap sort

```

1  type heap_struct = {data: int array; mutable size: int};;
2
3  let heap_sort t =
4    let max_size = Array.length t in
5    let heap = {data = Array.make (max_size + 1) 0; size = 0} in
6
7    let push x =
8      if heap.size = max_size then failwith "Heap overflow";
9
10     heap.data.(heap.size + 1) <- x;
11
12     let pos = ref (heap.size + 1) in
13     while !pos <> 1 do
14       let newpos = !pos / 2 in
15       if heap.data.(newpos) > heap.data.(!pos) then
16         swap !pos newpos heap.data;
17       pos := newpos;
18     done;
19
20     heap.size <- heap.size + 1
21   in
22
23   let pop () =
24     if heap.size = 0 then failwith "Empty heap";
25
26     let top = heap.data.(1) in
27     heap.data.(1) <- heap.data.(heap.size);
28
29     let changed = ref true in
30     let pos = ref 1 in
31     while !changed && 2 * !pos <= heap.size do
32       let minpos = ref !pos in
33       let left = 2 * !pos and right = 2 * !pos + 1 in
34       if heap.data.(!minpos) > heap.data.(left) then
35         minpos := left;
36       if right <= heap.size then
37         if heap.data.(!minpos) > heap.data.(right) then
38           minpos := right;
39
40       changed := not (!pos = !minpos);
41       swap !pos !minpos heap.data;
42
43       pos := !minpos;
44     done;
45
46     heap.size <- heap.size - 1;
47     top;
48   in
49
50   for pos = 0 to max_size - 1 do
51     push t.(pos)
52   done;
53
54   for pos = 0 to max_size - 1 do
55     t.(pos) <- pop ()
56   done;
57 ;;
```

4 Timing

We have run full measurements of these algorithms performance, which of course confirm the overall better performance of quicksort over other algorithms. We also timed the OCaml library `Array.sort` and `List.sort` functions; to our greatest surprise, it was largely outperformed by our simple, unoptimized functions, both when running top-level scripts and compiled code.

4.1 Methodology

We first define functions to generate random input data, for algorithms testing; then, we implement a timing function and, finally, we create a `time_fn` function which operates on a list of input sizes and generates tests for each input size.

```

1   let random_vect length =
2     let t = Array.make length 0 in
3     for pos = 0 to length - 1 do
4       t.(pos) <- Random.int max_int
5     done;
6     t;
7   ;;
8
9   let rec random_list = function
10    | 0 -> []
11    | n -> (Random.int max_int) :: (random_list (n-1))
12  ;;
13
14  let time randomize sort length =
15    let data = randomize length in
16    let start = Sys.time () in
17      sort data;
18      Sys.time () -. start;
19  ;;
20
21  let time_fn name random fn counts =
22    let rec _time_fn = function
23      | [] -> ()
24      | count :: tail ->
25          Printf.printf "%s : %d elements -> %f\n" name count (time random fn
26            count);
27          _time_fn tail
28    in _time_fn counts; print_newline ()
29  ;;

```

4.2 Testing our sorting algorithms

To test our algorithms, we ran the following commands:

```

1   time_fn "bubble_sort" random_vect bubble_sort [10; 100; 1000; 10000];
2   time_fn "insert_sort" random_vect insert_sort [10; 100; 1000; 10000];
3   time_fn "selection_sort" random_vect selection_sort [10; 100; 1000; 10000];
4   time_fn "quicksort" random_vect quicksort [10; 100; 1000; 10000; 100000; 1000000];
5   time_fn "merge_sort" random_vect merge_sort [10; 100; 1000; 10000; 100000; 1000000];
6   time_fn "heap_sort" random_vect heap_sort [10; 100; 1000; 10000; 100000; 1000000];
7   time_fn "Array.sort" random_vect (Array.sort compare) [10; 100; 1000; 10000; 100000; 1000000];
8
9   time_fn "r_bubble_sort" random_list r_bubble_sort [10; 100; 1000; 10000];
10  time_fn "r_insert_sort" random_list r_insert_sort [10; 100; 1000; 10000];
11  time_fn "r_selection_sort" random_list r_selection_sort [10; 100; 1000; 10000];
12  time_fn "r_selection_sort2" random_list r_selection_sort2 [10; 100; 1000; 10000];
13  time_fn "r_quicksort" random_list r_quicksort [10; 100; 1000; 10000; 100000; 1000000];
14  time_fn "r_merge_sort" random_list r_merge_sort [10; 100; 1000; 10000; 100000; 1000000];
15  time_fn "r_heap_sort" random_list r_heap_sort [10; 100; 1000; 10000; 100000; 1000000];
16  time_fn "List.sort" random_list (List.sort compare) [10; 100; 1000; 10000; 100000; 1000000];

```


4.3 Results

On a Intel Pentium 4 CPU, 3.20 GHz machine running Windows 7, we obtained the following results (all times are expressed in seconds):

```

bubble_sort : 10 elements -> 0.000000
bubble_sort : 100 elements -> 0.002000
bubble_sort : 1000 elements -> 0.234000
bubble_sort : 10000 elements -> 21.111000

insert_sort : 10 elements -> 0.000000
insert_sort : 100 elements -> 0.002000
insert_sort : 1000 elements -> 0.189000
insert_sort : 10000 elements -> 13.980000

selection_sort : 10 elements -> 0.000000
selection_sort : 100 elements -> 0.001000
selection_sort : 1000 elements -> 0.130000
selection_sort : 10000 elements -> 10.368000

quicksort : 10 elements -> 0.000000
quicksort : 100 elements -> 0.001000
quicksort : 1000 elements -> 0.006000
quicksort : 10000 elements -> 0.098000
quicksort : 100000 elements -> 0.939000
quicksort : 1000000 elements -> 11.571000

merge_sort : 10 elements -> 0.000000
merge_sort : 100 elements -> 0.000000
merge_sort : 1000 elements -> 0.007000
merge_sort : 10000 elements -> 0.113000
merge_sort : 100000 elements -> 1.104000
merge_sort : 1000000 elements -> 13.133000

heap_sort : 10 elements -> 0.000000
heap_sort : 100 elements -> 0.001000
heap_sort : 1000 elements -> 0.011000
heap_sort : 10000 elements -> 0.210000
heap_sort : 100000 elements -> 2.168000
heap_sort : 1000000 elements -> 28.454000

Array.sort : 10 elements -> 0.000000
Array.sort : 100 elements -> 0.000000
Array.sort : 1000 elements -> 0.008000
Array.sort : 10000 elements -> 0.097000
Array.sort : 100000 elements -> 1.221000
Array.sort : 1000000 elements -> 18.383000

r_bubble_sort : 10 elements -> 0.000000
r_bubble_sort : 100 elements -> 0.006000
r_bubble_sort : 1000 elements -> 0.431000
r_bubble_sort : 10000 elements -> 77.651000

r_insert_sort : 10 elements -> 0.000000
r_insert_sort : 100 elements -> 0.001000
r_insert_sort : 1000 elements -> 0.105000
r_insert_sort : 10000 elements -> 8.048000

r_selection_sort : 10 elements -> 0.000000
r_selection_sort : 100 elements -> 0.002000
r_selection_sort : 1000 elements -> 0.258000
r_selection_sort : 10000 elements -> 22.524000

r_selection_sort2 : 10 elements -> 0.000000
r_selection_sort2 : 100 elements -> 0.003000
r_selection_sort2 : 1000 elements -> 0.280000
r_selection_sort2 : 10000 elements -> 27.43800

r_quicksort : 10 elements -> 0.000000
r_quicksort : 100 elements -> 0.000000
r_quicksort : 1000 elements -> 0.004000
r_quicksort : 10000 elements -> 0.243000
r_quicksort : 100000 elements -> 1.011000
Stack overflow during evaluation (...).

r_merge_sort : 10 elements -> 0.000000
r_merge_sort : 100 elements -> 0.001000
r_merge_sort : 1000 elements -> 0.006000
r_merge_sort : 10000 elements -> 0.209000
r_merge_sort : 100000 elements -> 1.535000
Stack overflow during evaluation (...).

tr_merge_sort : 10 elements -> 0.000000
tr_merge_sort : 100 elements -> 0.002000
tr_merge_sort : 1000 elements -> 0.010000
tr_merge_sort : 10000 elements -> 0.151000
tr_merge_sort : 100000 elements -> 2.082000
Stack overflow during evaluation (...).

List.sort : 10 elements -> 0.000000
List.sort : 100 elements -> 0.000000
List.sort : 1000 elements -> 0.005000
List.sort : 10000 elements -> 0.145000
List.sort : 100000 elements -> 1.380000
Stack overflow during evaluation (...).

```

5 Conclusion

The library sorting functions implementation appears to be deficient.

Compared to the theoretic $O(n \lg(n))$, the library `Array.sort` function requires an approximate $36.8 \cdot n \lg(n)$ operations to complete, assuming an approximate $40 \cdot 10^6$ operations per second, while our own functions need $23.2 \cdot n \lg(n)$, $26.3 \cdot n \lg(n)$, and $57 \cdot n \lg(n)$ operations respectively. Heap sort, although being a very easy to implement algorithm, demonstrates bad performance, while both quicksort and merge sort outperform the OCaml library function `Array.sort` (by an approximate 36% for the former, and 28% for the later).

Similarly, recursive merge sort demonstrates disappointing performance ($37 \cdot n \lg(n)$), whereas recursive quicksort achieves overall good performance ($24.3 \cdot n \lg(n)$), outperforming the library `List.sort` function ($33.2 \cdot n \lg(n)$) by an approximate 26%.

Identical results were obtained when running compiled versions of these functions. Note that due to stack overflow exceptions – occurring as a result of using non-tail-recursive algorithms –, constants were evaluated based on 10^6 elements sets in the array sorting case, whereas 10^5 elements sets were used in the list sorting case.